
Messaging Functions

Functions Synopsis

FOCUS.FLL provides a basic messaging mechanism. It is working on the Post Office scheme, which basically means that all messages are stored in a common place: the Post Office!

The Post Office is actually 2 things :

1. it is a common place in which all messages are stored

2. it is a Settings File in which all accounts are defined.

For different applications to exchange messages, they all have to agree on a common Post office and a common Settings File. This is done through the MSG_PostOffice() function. For example, by issuing the following command, many different applications specify they rely on the Post Office located on the WALK2227 machine, in the Post Office directory, and that the Settings File to take in consideration is the PO.INI :

MSG_PostOffice( "\\WALK2227\Post Office\PO.INI" )

Each application should be uniquely identified. In our example, we have 7 different applications running on different machines but they all access one common central Post Office and they all use the same Settings File.

Structure of the Settings File

[MESSAGES]
Counter=111

[SEQUENCING]
next=111

[ACCOUNTS]
APPLI1=00000001
APPLI2=00000002
APPLI3=00000003
<etc.>

[DEADLETTER]
Dir=Dead

[APPLI1]
Dir=00000001
Comment=Elected PC

[APPLI]
Dir=00000002
Comment=mailbox of COMPAQ 166 MHz

[APPLI3]
Dir=00000003
Comment=mailbox of Dominique's PC

Post Office

Technically, the Post Office is nothing else but a shared directory on a network. Each application (recipient) is uniquely identified and is granted its own directory within the Post Office. For example, on the WALK2227 machine, the central Post office is located in the Post Office directory, and Appli1 is hooked to the 00000001 directory within the Post Office; Appli2 is hooked to the 00000002 directory, and so on (according to the structure of the Settings File).

How To Work With Messaging Functions?

Imagine the following situation : you have 3 applications, Appli1, Appli2 and Appli3. Appli1 and Appli2 are located on the MACHINE1 computer while Appli3 is placed on MACHINE2. You want the Post Office to be located on a third machine, MACHINE3, in the BP directory. The first thing to do is to create a Post Office Settings File similar to:

[MESSAGES]
Counter=0

[SEQUENCING]
next=0

[ACCOUNTS]
APPLI1=00000001
APPLI2=00000002
APPLI3=00000003

[DEADLETTER]
Dir=Dead

[APPLI1]
Dir=00000001

[APPLI]
Dir=00000002

[APPLI3]
Dir=00000003

You place this Settings File in the BP directory of the Post Office (\\MACHINE3\BP, a directory that you share for both Read and Write operations) and you name it PO.INI. On MACHINE3, directory BP, you will need to create the subdirectories that each application is hooked to, in other words 00000001, 00000002, and 00000003.

Now, you're ready to use Messaging Functions of FOCUS.FLL. For example, the following code that gets executed in Appli1 will broadcast a message to the other two applications that are served by the same Post Office:

MSG_PostOffice( "\\MACHINE3\BP\PO.INI" )

IF ( ! MSG_Broadcast( "APPLI1","Subject","Category","This is my message" ) )
   ? "Not all recipients were served:",MSG_LastError()
ENDIF

As you can see, there is no mentioning of the machine an application is on for the Messaging Functions to work without any problem.

Now one might ask : How do I know when a message has arrived? The answer is clear and obvious (for one that already pointed a head in the direction of the Notification Functions) : you simply set a notification hook.

For example, Appli3 might want to monitor its own directory on the Post office to be warned whenever a new message arrives. Therefore it simply needs to use the following code :

LOCAL nHandle
LOCAL lSubDir

#define MONITOR_LAST_WRITE   0x00000010

lSubDir = .T.

MSG_PostOffice( "\\MACHINE3\BP\PO.INI" )            && Specify which Post office

nHandle = NOT_set( "\\MACHINE3\BP\00000003", ;      && Directory to watch
                   "MyMsgHandler"          , ;      && Proc called in case of message
                   lSubDir                 , ;      && Watch also sub-directories
                   MONITOR_LAST_WRITE )             && Monitor last write changes

&& If invalid handle
IF ( nHandle == -1 )
   ? "Cannot set notification on this directory"
ELSE
    NOT_frequency(1000)     && Allows a delay of 1 second (1000 milliseconds)
ENDIF

PROCEDURE MyMsgHandler( n )
    LOCAL cMessage

    **********************************************************************
    * PURPOSE : Whenever thios procedure is called it means that there is
    *           a new message coming in the mailbox of this application
    **********************************************************************

    && Read the least recent message of this application
    cMessage = MSG_read( "APPLI3" )

    && If we successfully read the message, display it
    IF ( ! EMPTY( cMessage ) )
       ? cMessage
    ENDIF

RETURN

An application can dynamically create multiple Post Offices although one and only one Post Office is the current Post Office. To do so, it must call the MSG_CreatePostOffice(), and MSG_CreateAccount() functions of FOCUS.FLL. This allows intelligent applications to dialog between each other beyond the knowledge of any user.

FOCUS.FLL provides also some statistical functions (although they are very few) to tighter control messages : MSG_TotalMessagesSent(), MSG_BytesSent(), MSG_MessagesSent(), MSG_MessagesRead(), MSG_MessagesReceived().

MSG_Broadcast() : Sends a message to all recipients of the Post Office.

Syntax

MSG_Broadcast( szFrom,szSubject,szCategory,szBody )  lSuccess

Parameters

szFrom
sender identification.

szSubject
subject of the message.

szCategory
category of the message.

szBody
body of the message.

Returns

lSuccess
.T. if message was successfully sent; .F. if not. There is no message sent to the sender.

Example

MSG_PostOffice( "\\WALK2227\Post Office\PO.INI" )
IF ( ! MSG_Broadcast( "ME","System","Life","ME still alive" ) )
   ? "Message cannot be sent",MSG_LastError()
ENDIF

MSG_BytesSent() : Total number of bytes sent with the current Post Office.

Syntax

MSG_BytesSent()  nBytes

Parameters

None.

Returns

nBytes
number of bytes sent. Only the body of a message is taken in account.

Example

? "Average message length:",MSG_BytesSent()/MSG_TotalMessagesSent()

MSG_count() : Counts the number of unread messages for a given recipient.

Alias

MSG_MessagesUnread()

Syntax

MSG_count( szRecipient )  nMessages

Parameters

szRecipient
recipient identification.

Returns

nMessages
the number of waiting messages or -1 if the recipient is unknown.

Example

? MSG_PostOffice( "C:\PO.INI" )
? MSG_count( "APPLI5" )                                               && 0
? MSG_count( "UNKNOWN RECIPIENT" )                                    && -1
? MSG_send( "APPLI1","APPLI5","Subject","Category",TTOC(DATETIME()) ) && .T.
? MSG_count( "APPLI5" )                                               && 1
? MSG_send( "APPLI1","APPLI5","Subject","Category",TTOC(DATETIME()) ) && .T.
? MSG_count( "APPLI5" )                                               && 2

MSG_CreateAccount() : Creates an account in the current Post Office.

Syntax

MSG_CreateAccount( szAccount,szDir )  lSuccess

Parameters

szAccount
account identification.

szDir
directory in which messages for szAccount will be saved.

Returns

lSuccess
.T. if the account was successfully created; .F. if not.

Example

IF ( MSG_CreatePostOffice( "\\WALK2227\Post Office\PO.INI" ) )
    IF ( MSG_CreateAccount( "Account1","00000001" ) )
       ? "Account successfully recognized"
    ENDIF
ELSE
   ? "Post Office cannot be created"
ENDIF

MSG_CreatePostOffice() : Creates a Post Office.

Syntax

MSG_CreatePostOffice( szPO )  lSuccess

Parameters

szPO
Post Office Settings file and location.

Returns

lSuccess
.T. if the Post Office was successfully created; .F. if not. The current Post Office is changed to szPO.

Example

IF ( MSG_CreatePostOffice( "\\WALK2227\Post Office\PO.INI" ) )
    IF ( MSG_CreateAccount( "Account1","00000001" ) )
       ? "Account successfully recognized"
    ENDIF
ELSE
   ? "Post Office cannot be created"
ENDIF

MSG_DeleteAccount() : Deletes an account in the current Post Office.

Syntax

MSG_DeleteAccount( szAccount,lSubdir )  lSuccess

Parameters

szAccount
account identification.

lSubdir
deletes the directory in which messages of this account have been stored including all messages that are still to be read.

Returns

lSuccess
.T. if the account was successfully deleted; .F. if not. If the return value is .F., call MSG_LastError() to get a more explicit reason.

MSG_DeleteUnread() : Deletes all pending messages of an account.

Alias

MSG_DeleteAllMessages(), MSG_DeleteMessages(), DeleteMessages(), DeleteAllMessages()

Syntax

MSG_DeleteUnread( szAccount )  nDeleted

Parameters

szAccount
account identification.

Returns

nDeleted
number of messages actually deleted or –1 if the account is not found or the current Post Office not set.

Example

* Simulates a right click where the mouse is 
MOU_RightDown()
MOU_RightUp()

MSG_GetAllAccounts() : Gets all accounts that are defined in the current Post Office.

Alias

MSG_GetAccounts(), MSG_Accounts()

Syntax

MSG_GetAllAccounts()  szAccounts

Parameters

None.

Returns

szAccounts
a tokenized string with all accounts defined in the current Post Office. Each token is separated by a ";".

Example

LOCAL nAccounts
LOCAL OldSep
LOCAL i
LOCAL szAccounts

m.OldSep = STR_setsep( ";" )

MSG_PostOffice( "\\MYSERVER\Bureau\BP.INI" )

IF ( MSG_IsPostOfficeAccessible() )

    m.szAccounts = MSG_GetAllAccounts()

    IF ( ! EMPTY( m.szAccounts ) )

        m.nAccounts = STR_numtok( m.szAccounts )

        FOR i = 1 TO m.nAccounts
           ? "Account : " + STR_ntoken( i,m.szAccounts )
        NEXT

    ENDIF

ENDIF

STR_setsep( m.OldSep )

MSG_GetDir() : Returns the mailbox directory of a given recipient (absolute path).

Syntax

MSG_GetDir( szRecipient )  szDir

Parameters

szRecipient
recipient identification.

Returns

szDir
the mailbox directory or an empty string is the mailbox directory cannot be found in the Settings File. When an empty string is returned, use the MSG_LastError() function to determine the cause of the problem.

Example

? MSG_PostOffice("C:\PO.INI")
? MSG_GetDir("APPLI5")       && "C:\00000005"

See also

MSG_GetSubDir()
MSG_GetSep() : Returns the separator that is used internally by FOCUS.FLL to tokenize a message.

Syntax

MSG_GetSep()  szSep

Parameters

None.

Returns

szSep
the separator used internally by FOCUS.FLL to tokenize a message.

Example

LOCAL cMessage

LOCAL OldSep
LOCAL szFrom
LOCAL szTo
LOCAL szDateTime
LOCAL szSubject
LOCAL szCategory
LOCAL szBody
LOCAL nTokens

MSG_PostOffice( "\\WALK2227\Post Office\PO.INI" )

szMessage = MSG_Read( "ME" )

IF ( EMPTY( szMessage ) )
   ? "Empty message"
ELSE
    OldSep  = STR_setsep( MSG_GetSep() )
    nTokens = STR_numtok( szMessage )
    IF ( nTokens == 6 )
        szFrom     = STR_ntoken( 1,szMessage )
        szTo       = STR_ntoken( 2,szMessage )
        szDateTime = STR_ntoken( 3,szMessage )
        szSubject  = STR_ntoken( 4,szMessage )
        szCategory = STR_ntoken( 5,szMessage )
        szBody     = STR_ntoken( 6,szMessage )
    ELSE
       ? "Unexpected number of tokens"
    ENDIF
    STR_setsep( OldSep )
ENDIF

MSG_GetSubDir() : Returns the mailbox directory of a given recipient (relative path).

Syntax

MSG_GetSubDir( szRecipient )  szDir

Parameters

szRecipient
recipient identification.

Returns

szDir
the mailbox directory or an empty string is the mailbox directory cannot be found in the Settings File. When an empty string is returned, use the MSG_LastError() function to determine the cause of the problem.

Example

? MSG_PostOffice("C:\PO.INI")
? MSG_GetSubDir("APPLI5")       && "00000005"

See also

MSG_GetDir()
MSG_IsAccount() : Determines if the specified account is known bt the current Post Office.

Syntax

MSG_IsAccount( szAccount )  lExist

Parameters

szAccount
account identification.

Returns

lExist
.T. if szAccount exists in current Post Office; .F. if not.

Example

MSG_PostOffice( "C:\PO.INI" )
IF ( MSG_IsPostOfficeAccessible() )
    IF ( MSG_IsAccount( "SYSTEM" ) )
        IF ( MSG_send( "ME","SYSTEM","Test","Category","My message" )
           ? "SYSTEM has received",MSG_MessagesReceived( "SYSTEM" ),"messages"
        ENDIF
    ENDIF
ENDIF

MSG_IsPostOfficeAccessible() : Determines whether the Post Office can be contacted or not.

Alias

MSG_IsPostOffice(), MSG_IsPO(), MSG_IsPOAccessible(), IsPostOffice(), IsPostOfficeAccessible(), IsPO()

Syntax

MSG_IsPostOfficeAccessible()  lSuccess

Parameters

None.

Returns

lSuccess
.T. if the Post Office can be contacted; .F. if not.

Example

LOCAL nAccounts
LOCAL OldSep
LOCAL i
LOCAL szAccounts

OldSep = STR_setsep( ";" )

MSG_PostOffice( "\\WALK2227\Bureau\BP.INI" )

IF ( MSG_IsPostOfficeAccessible() )

    szAccounts = MSG_GetAllAccounts()

    IF ( ! EMPTY( szAccounts ) )

        nAccounts = STR_numtok( szAccounts )

        FOR i = 1 TO nAccounts
           ? "Account : " + STR_ntoken( i,szAccounts )
        NEXT

    ENDIF

ENDIF

STR_setsep( OldSep )

MSG_LastError() : Returns a string indicating the last error in Messaging functions.

Alias

MSG_GetLastError()

Syntax

MSG_LastError()  szError

Parameters

None.

Returns

szError
message describing the last error that occurred in messaging functions.

Example

LOCAl nMessages

? MSG_PostOffice("C:\PO.INI")


nMessages = MSG_count( "UNKNOWN RECIPIENT" )


IF ( nMessages == -1 )
   ? MSG_LastError()           && "Cannot find recipient"
ENDIF

MSG_LastVersion() : Returns the file stamp of MSG functions.

Remark

This function helps the developer identifying the last version of a set of functions. Sometimes the global version information of FOCUS.FLL (MIS_major() and MIS_minor()) does not help tracking down the changes in a project. Starting with version 6.0 of FOCUS.FLL, each source file has now an internal date and time stamp.

Syntax

MSG_LastVersion()  szLastVersion

Parameters

None.

Returns

szLastVersion
string identifying the last version of the functions set. The string is similar to "C:\Focus\5.0\MSG.C-Mon Oct 19 15:55:22 1998".

MSG_MessagesRead() : Returns the number of messages the specified recipient has already read.

Syntax

MSG_MessagesRead( szRecipient )  nMessages

Parameters

szRecipient
recipient identification.

Returns

nMessages
number of messages.

MSG_MessagesReceived() : Returns the number of messages the specified recipient has received in total.

Syntax

MSG_MessagesReceived( szRecipient )  nMessages

Parameters

szRecipient
recipient identification.

Returns

nMessages
number of messages.

MSG_MessagesSent() : Returns the number of messages a specified originator has sent.

Syntax

MSG_MessagesSent( szOriginator )  nMessages

Parameters

szOriginator
originator identification.

Returns

nMessages
number of messages.

MSG_PO() : Gets the Post Office location.

Syntax

MSG_PO()  szPOLoc

Parameters

None.

Returns

szPOLoc
current Post Office location.

Example

MSG_PostOffice( "\\WALK2227\Post Office\PO.INI" )
? MSG_PO()      && "\\WALK2227\Post Office"
? MSG_POFile()  && "PO.INI"

MSG_POFile() : Gets the Post Office settings file.

Syntax

MSG_POFile()  szPOFile

Parameters

None.

Returns

szFile
current Post Office settings file.

Example

MSG_PostOffice( "\\WALK2227\Post Office\PO.INI" )
? MSG_PO()      && "\\WALK2227\Post Office"
? MSG_POFile()  && "PO.INI"

MSG_PostOffice() : Gets/Sets the Post office location and settings file.

Alias

PostOffice()

Syntax

MSG_PostOffice( szPO ) szCurrentPO

Parameters

sz
Post Office location and settings file. The settings file is typically an INI file.

Returns

szCurrentPO
current Post office location and settings file.

Example

MSG_PostOffice( "\\WALK2227\Post Office\PO.INI" )
? MSG_PO()      && "\\WALK2227\Post Office"
? MSG_POFile()  && "PO.INI"

MSG_Read() : Reads the least recent message of a given recipient.

Syntax

MSG_Read( szRecipient[,lDelete] )  cMessage

Parameters

szRecipient
recipient identification.

lDelete
should the message be deleted? By default, the message is deleted from the Post office.

Returns

szMessage
the message string is composed of 6 tokens. Each token is separated by a CHR(1). The separator can be queried by the programmer by the MSG_GetSep() function.

Example

LOCAL szMessage

LOCAL OldSep
LOCAL szFrom
LOCAL szTo
LOCAL szDateTime
LOCAL szSubject
LOCAL szCategory
LOCAL szBody
LOCAL nTokens

MSG_PostOffice( "\\WALK2227\Post Office\PO.INI" )

szMessage = MSG_Read( "ME" )

IF ( EMPTY( szMessage ) )
   ? "Empty message"
ELSE
    OldSep  = STR_setsep( MSG_GetSep() )
    nTokens = STR_numtok( szMessage )
    IF ( nTokens == 6 )
        szFrom     = STR_ntoken( 1,szMessage )
        szTo       = STR_ntoken( 2,szMessage )
        szDateTime = STR_ntoken( 3,szMessage )
        szSubject  = STR_ntoken( 4,szMessage )
        szCategory = STR_ntoken( 5,szMessage )
        szBody     = STR_ntoken( 6,szMessage )
    ELSE
       ? "Unexpected number of tokens"
    ENDIF
    STR_setsep( OldSep )
ENDIF

MSG_Send() : Sends a message.

Syntax

MSG_Send( szFrom,szTo,szSubject,szCategory,szBody )  lSuccess

Parameters

szFrom
sender identification.

szTo
recipient identification.

szSubject
subject of the message.

szCategory
category of the message.

szBody
body of the message.

Returns

lSuccess
.T. if message was sent successfully; .F. if not.

Example

MSG_PostOffice( "\\WALK2227\Post Office\PO.INI" )
IF ( ! MSG_Send( "ME","YOU","Subject","Topic","This is my message" ) )
   ? "Message cannot be sent",MSG_LastError()
ENDIF

MSG_SetPONotification() : Sets a directory notification on a specified directory.

Syntax

MSG_SetPONotification( szDir,szProc )  lSuccess

Parameters

szDir
directory to monitor for messages.

szProc
procedure to execute whenever a messages arrives in szDir.

Returns

lSuccess
.T. if directory notification mechanism successfully hooked to szDir; .F. if not.

Example

MSG_Successful() : Returns the string that is used to declare that the last operation has been successfully executed.

Syntax

MSG_Successful()  szConstant

Parameters

None.

Returns

szConstant
"The command completed successfully".

Example

&& Call whatever MSG function, and then simply test whether
&& MSG_LastError() is identical to MSG_Successful()

MSG_TotalMessagesSent() : Returns the total number of messages sent through the current Post Office.

Syntax

MSG_TotalMessagesSent()  nMessages

Parameters

None.

Returns

nMessages
number of messages sent.

Example

? "Average message length:",MSG_BytesSent()/MSG_TotalMessagesSent()






